APPENDIX B

Some special functions in low-frequency seismology

1. Spherical Bessel functions

The functions j_l and y_l are useful in constructing mode solutions in homogeneous spheres (fig B1). They satisfy

$$\frac{d^2 j_l}{dr^2} + \frac{2}{r}\frac{dj_l}{dr} + \left[k^2 - \frac{l(l+1)}{r^2}\right] j_l = 0 \tag{B.1}$$

where $j_l(x)$ has the argument $x = kr$. Recurrence relations for j_l and y_l are given by Abramowitz and Stegun:

$$j_{l-1} + j_{l+1} = \frac{2l+1}{x} j_l \tag{B.2}$$

$$j_l' = j_{l-1} - \frac{l+1}{x} j_l \tag{B.3}$$

$$j_l' = \frac{l}{x} j_l - j_{l+1} \tag{B.4}$$

where prime denotes differentiation with respect to x.



Figure B1. Spherical Bessel functions j_l and y_l for $l = 0, 1, 2, 3, 4$.s.

Upward recursion using B2 is stable for y_l but is unstable for j_l. Downward recursion (Miller's algorithm) can be used for j_l and is described in Press *et al.*, (1986). The ratios j_l/j_{l-1} are well-behaved and can be computed using a continued fraction algorithm (Lentz, 1976) which is shown in fig B.2. This algorithm also works for complex order and argument.

```
   subroutine sphbr(j,y,x,l)
Calculate spherical Bessel functions of the 1st and 2nd kind with real
arguments.
 on input
 x = arg of Bessel function (not zero)
 l = max angular order to be computed
 on output.
 j(l+1) contains the spherical Bessel function of order l
 y(l+1) contains the spherical Bessel function of order l
Continued fraction technique for computing spherical Bessel functions in
Mie scattering, see : W. Lentz ,Applied Optics, vol.15, #3, March, 1976
   implicit real*8(a-h,o-z)
   real*8 nu,numer,j(1),y(1)
   data tol/1.d-14/
   lp1=l+1
   rxsq=1.d0/(x*x)
   j(1)=sin(x)/x
   y(1)=-cos(x)/x
   do n=2,lp1
     rx=2.d0/x
     nu=n-0.5d0
     ans=nu*rx
     nu=nu+1.d0
     rx=-rx
     denom=nu*rx
     numer=denom+1.d0/ans
 5   ratio=numer/denom
     ans=ans*ratio
     if(abs(abs(ratio)-1.d0).gt.tol) then
       nu=nu+1.d0
       rx=-rx
       a=nu*rx
       denom=a+1.d0/denom
       numer=a+1.d0/numer
       goto 5
     end if
     j(n)=j(n-1)/ans
     y(n)=y(n-1)/ans-rxsq/j(n-1)
   enddo
   return
   end
```

Figure B2. A Fortran program to compute spherical bessel functions.

Some values for small l are:

l	j_l	y_l
0	$\sin x/x$	$-\cos x/x$
1	$\sin x/x^2 - \cos x/x$	$-\cos x/x^2 - \sin x/x$
2	$(3/x^3 - 1/x)\sin x - 3\cos x/x^2$	$(-3/x^3 + 1/x)\cos x - 3\sin x/x^2$

For small values of x:

$$\left.\begin{aligned} j_l(x) =& \frac{x^l}{1\cdot 3\cdot 5\cdots(2l+1)}\left[1-\frac{x^2/2}{(2l+3)}+\cdots\right. \\ y_l(x) =& \frac{1\cdot 3\cdot 5\cdots(2l-1)}{x^{l+1}}\left[1+\frac{x^2/2}{(2l-1)}+\cdots\right. \end{aligned}\right\} \tag{B.5}$$

For large values of x:

$$\left.\begin{aligned} j_l(x) \simeq& \frac{1}{x}\sin(x-l\pi/2) \\ y_l(x) \simeq& -\frac{1}{x}\cos(x-l\pi/2) \end{aligned}\right\} \tag{B.6}$$

2. Surface spherical harmonics

The surface spherical harmonics are given by:

$$Y_l^m(\theta,\phi) = X_l^m(\theta)e^{im\phi} \tag{B.7}$$

$$\text{where}\quad X_l^m(\theta) = (-1)^m\left[\frac{2l+1}{4\pi}\frac{(l-m)!}{(l+m)!}\right]^{\frac{1}{2}} P_l^m(\cos\theta) \tag{B.8}$$

Note that

$$Y_l^{-m} = (-1)^m Y_l^{m\star} \tag{B.9}$$

so we only need compute these functions for non-negative m. The spherical harmonics satisfy

$$\nabla_1^2 Y_l^m = -l(l+1)Y_l^m \tag{B.10}$$

$$\text{where}\quad \nabla_1^2 = \frac{\partial^2}{\partial\theta^2} + \cot\theta\frac{\partial}{\partial\theta} + \operatorname{cosec}^2\theta\frac{\partial^2}{\partial\phi^2}$$

This last equation allows us to easily evaluate any θ derivative of Y_l^m provided we can compute $\partial Y_l^m/\partial\theta$. The P_l^m's in B8 are the associated Legendre functions which are defined by:

$$P_l^m(x) = \frac{(1-x^2)^{m/2}}{2^l l!}\frac{d^{l+m}}{dx^{l+m}}(x^2-1)^l = (1-x^2)^{m/2}\frac{d^m P_l(x)}{dx^m} \tag{B.11}$$

$$\text{and}\quad P_l^{-m}(x) = (-1)^m\frac{(l-m)!}{(l+m)!}P_l^m(x) \tag{B.12}$$

They satisfy various recurrence relations:

$$(l-m+1)P_{l+1}^m - (2l+1)xP_l^m + (l+m)P_{l-1}^m = 0 \tag{B.13}$$

$$(1-x^2)^{\frac{1}{2}}P_l^{m+1} - 2mxP_l^m + (l+m)(l-m+1)(1-x^2)^{\frac{1}{2}}P_l^{m-1} = 0 \tag{B.14}$$

$$(1-x^2)\frac{dP_l^m}{dx} = (l+1)xP_l^m - (l-m+1)P_{l+1}^m \tag{B.15}$$

$$(1-x^2)\frac{dP_l^m}{dx} = (l+m)P_{l-1}^m - lxP_l^m \tag{B.16}$$

and the governing equation is

$$(1-x^2)\frac{d^2P_l^m}{dx^2} - 2x\frac{dP_l^m}{dx} + \left[l(l+1) - \frac{m^2}{1-x^2}\right]P_l^m = 0 \tag{B.17}$$

An extremely stable way of computing the X_l^m's is to use the l recurrence B13 starting at $l=0$ and then use B16 to compute the θ derivative (remember $x = \cos\theta$). Because $P_l^m = 0$ if $m > l$, all we need to implement the algorithm is a starting value for P_l^l which is given by:

$$P_l^l = (\sin\theta)^l(2l-1)!! \tag{B.18}$$

$$\text{where} \quad (2l-1)!! = 1\cdot 3\cdot 5\cdots(2l-1) = \frac{(2l)!}{2^l l!}$$

This method is extremely inefficient if we are only interested in Y_l^m for a single harmonic degree. If we are careful, we can use B14 to recur over m. This recurrence is stable only if we start the recurrence at $m=l$ and go to $m=0$. We rewrite the recurrence in terms of the X_l^m's:

$$\left.\begin{aligned} X_l^{m-1} &= -\left[\frac{dX_l^m}{d\theta} + m\cot\theta X_l^m\right]/[(l+m)(l-m+1)]^{\frac{1}{2}} \\ \text{and} \quad \frac{d}{d\theta}X_l^{m-1} &= (m-1)\cot\theta X_l^{m-1} + X_l^m[(l+m)(l-m+1)]^{\frac{1}{2}} \end{aligned}\right\} \tag{B.19}$$

with the starting values

$$\left.\begin{aligned} X_l^l &= (-1)^l\left[\frac{2l+1}{4\pi}\frac{1}{(2l)!}\right]^{\frac{1}{2}}\frac{(2l)!}{2^l l!}(\sin\theta)^l \\ \text{and} \quad \frac{d}{d\theta}X_l^l &= l\cot\theta X_l^l \end{aligned}\right\} \tag{B.20}$$

For large l and small θ, the starting value, X_l^l, can underflow (be indistinguishable from zero) on the computer and so we must re-scale or resort to l recursions. A program implementing this algorithm is given in fig B.3 and more detail can be found in the appended reprint of Masters and Richards-Dinger.

We sometimes need to know the behavior of Y_l^m and its derivatives at the origin (as $\theta,\phi\to 0$). This is usually most easily done by recasting in terms of generalized spherical harmonics (see appendix A) but for reference we give the following result. Let $\theta=\epsilon$ and define

$$\mathrm{b}_l^m = \frac{(-1)^m}{2^m m!}\left[\frac{2l+1}{4\pi}\frac{(l+m)!}{(l-m)!}\right]^{\frac{1}{2}}$$

then

$$X_l^m = \mathrm{b}_l^m\epsilon^m[1 - A_l^m\epsilon^2 + O(\epsilon^4) \quad \text{where} \quad A_l^m = \frac{3l(l+1)-m(m+1)}{12(m+1)} \tag{B.21}$$

Finally, we sometimes need to know the behavior for large l. The asymptotic expansion for P_l^m valid when $l \gg 1/\epsilon$, $l\gg m$ and $\epsilon\le\theta\le\pi-\epsilon$ is

```
      subroutine xfcn(l,c,s,x,x1)
c  computes ordinary spherical harmonics (Edmonds eqn 2.5.29)
c    x(l,m,theta)*exp(imphi) are a set of orthonormal spherical
c    harmonics where theta is colatitude and phi is longitude.
c  input:
c    l=harmonic degree
c    c=cos(theta)
c    s=sin(theta)
c  output:
c    x(1) contains m=0, x(2) contains m=1, ... x(l+1) contains m=l
c    where m=azimuthal order 0.le.m.le.l .
c    x1=dx/dtheta stored in the same way as x
c  calls no other routines
c        TGM
      implicit real*8(a-h,o-z)
c*** note that harmonic degree must be .le. parameter lmx
      parameter (lmx=1000)
      dimension x(*),x1(*),iexp(lmx+1),scl(6)
      data pi4/12.56637061435916d0/
      data scl/1.d0,1.d-60,1.d-120,1.d-180,1.d-240,1.d-300/
      if(l.gt.lmx) then
         stop 'lmx in xfcn needs increasing'
      end if
      lp1=l+1
      fl2p1=l+lp1
      con=sqrt(fl2p1/pi4)
c*** handle special case of l=0
      if(l.eq.0) then
        x(1)=con
        x1(1)=0.d0
        return
      end if
      if(dabs(s).lt.1.d-20) then
c*** handle very small arguments
        do i=1,lp1
          x(i)=0.d0
          x1(i)=0.d0
        enddo
        x(1)=con
        x1(2)=-0.5d0*con*sqrt(dble(l*lp1))
        return
      end if
      cot=c/s
c*** first compute x_l^l
      f=1.d0
      do i=1,l
        f=f*(i+i-1.d0)/(i+i)
      enddo
      x(lp1)=con*dsqrt(f)*(-s)**l
      if(dabs(x(lp1)).gt.1.d-295) then
c*** use m recurrence starting from m=l, no scaling needed
        x1(lp1)=x(lp1)*l*cot
        do i=1,l
          m=lp1-i
          mp1=m+1
          f=sqrt(i*(fl2p1-i))
          x(m)=-(x1(mp1)+m*cot*x(mp1))/f
          x1(m)=(m-1)*x(m)*cot+x(mp1)*f
        enddo
      else
c*** use m recurrence starting from m=l with scaling
        x(lp1)=(-1.d0)**l
        x1(lp1)=x(lp1)*l*cot
        iexp(lp1)=0
        do i=1,l
          m=lp1-i
          mp1=m+1
          iexp(m)=iexp(mp1)
          f=sqrt(i*(fl2p1-i))
          x(m)=-(x1(mp1)+m*cot*x(mp1))/f
          x1(m)=(m-1)*x(m)*cot+x(mp1)*f
  100     if(abs(x1(m)).gt. 1.d60) then
            iexp(m)=iexp(m)+1
            x(m)=x(m)*1.d-60
            x1(m)=x1(m)*1.d-60
            goto 100
          end if
        enddo
c*** unexponentiate and calculate sum (addition rule)
        sum=x(1)*x(1)
        do i=2,lp1
          iex=iexp(1)-iexp(i)+1
          if(iex.lt.7) then
            x(i)=x(i)*scl(iex)
            x1(i)=x1(i)*scl(iex)
            sum=sum+2.d0*x(i)*x(i)
          else
            x(i)=0.d0
            x1(i)=0.d0
          end if
        enddo
        sum=con/dsqrt(sum)
c*** normalize using addition rule
        do i=1,lp1
          x(i)=x(i)*sum
          x1(i)=x1(i)*sum
          if(x(i).eq.0.d0) x1(i)=0.d0
        enddo
      end if
      return
      end
```

Figure B3. A Fortran program to compute spherical harmonics.

$$P_l^m(\cos\theta) \simeq (-l)^m \left(\frac{2}{\pi l \sin\theta}\right)^{\frac{1}{2}} \cos\left[(l+\tfrac{1}{2})\theta - \frac{\pi}{4} + \frac{m\pi}{2}\right] \tag{B.22}$$

More commonly, we need the large l behavior for the X_l^m which, correct to $O(l^{-2})$ is

$$X_l^m(\theta) = \frac{1}{\pi\sqrt{\sin\theta}} \cos\left[(l+\tfrac{1}{2})\theta - \frac{\pi}{4} + \frac{m\pi}{2} + (m^2 - 1/4)\frac{\cot\theta}{2l+1}\right] \tag{B.23}$$

Fig B.4 shows the behavior of X_{100}^m for various values of θ. The cosinusoidal behavior for small m is quite pronounced.



Figure B4. X_{100}^m as a function of m for various values of θ. Note the change from oscillatory to exponential behavior in the top two plots.

We shall also encounter integrals of combinations of spherical harmonics over the unit sphere. Some integrals of the product of teo spherical harmonics are:

$$\int\limits_{\Omega} Y_l^{m'^*} Y_l^m \, d\Omega = \delta_{mm'} \tag{B.24}$$

$$\int\limits_{\Omega} \text{cosec}^2\theta Y_l^{m'^*} Y_l^m \, d\Omega = \frac{2l+1}{2m} \delta_{mm'} \tag{B.25}$$

$$\int\limits_{\Omega} \frac{dY_l^{m'^*}}{d\theta} \frac{dY_l^m}{d\theta} \, d\Omega = \left[l(l+1) - \frac{m(2l+1)}{2} \right] \delta_{mm'} \tag{B.26}$$

$$\int\limits_{\Omega} \cot\theta Y_l^{m'^*} \frac{dY_l^m}{d\theta} \, d\Omega = \frac{1}{2} \delta_{mm'} \tag{B.27}$$

where $d\Omega = \sin\theta d\theta d\phi$.

Also useful is the integral around a great-circle path (Backus 1964 BSSA 54, 571 – 610):

$$\frac{1}{2\pi} \oint\limits_{\Theta,\Phi} Y_s^t(\theta,\phi) \, d\Delta = P_s(0) Y_s^t(\Theta,\Phi) \tag{B.28}$$

where Θ, Φ is the (positive) pole of the great-circle.

RESEARCH NOTE

On the efficient calculation of ordinary and generalized spherical harmonics

Guy Masters and Keith Richards-Dinger
IGPP, Scripps Institution of Oceanography, UCSD, La Jolla, CA 92093-0225, *USA. E-mail: guy@mahi.ucsd.edu*

Accepted 1998 April 20. Received 1998 March 31; in original form 1998 January 19

SUMMARY

Algorithms for the stable computation of generalized and ordinary spherical harmonics are presented. The algorithms are fast and have the useful property that they can compute harmonics for isolated harmonic degrees. FORTRAN and C programs implementing these algorithms are available from the authors.

Key words: generalized spherical harmonics, ordinary spherical harmonics.

ORDINARY SPHERICAL HARMONICS

Spherical harmonics are ubiquitous in geophysics, and many algorithms are available for their computation. Generally, all spherical harmonics up to some maximum harmonic degree are needed and many algorithms are available to do this (e.g. Press *et al.* 1992). In some applications, only a single harmonic degree is needed, and the purpose of this note is to give fast and accurate algorithms for this situation. The spherical harmonics in most common use in the seismology literature, and the ones we shall consider here, are those defined by Edmonds (1960); that is,

$$Y_l^m(\theta, \phi) = X_l^m(\theta)\, e^{im\phi}, \tag{1}$$

where

$$X_l^m(\theta) = (-1)^m \left[\frac{2l+1}{4\pi} \frac{(l-m)!}{(l+m)!} \right]^{1/2} P_l^m(\cos\theta), \tag{2}$$

where the P_l^m are associated Legendre functions, θ is colatitude and ϕ is longitude. This definition gives Y_l^ms that are fully normalized in the sense

$$\int_S Y_{l'}^{m'*}\, Y_l^m\, dS = \delta_{mm'}\delta_{ll'},$$

where

$$dS = \sin\theta\, d\theta\, d\phi\,.$$

We often need to compute Y_l^ms and their derivatives with respect to θ. The following algorithms are sufficiently fast that they can be used to calculate all Y_l^ms up to some maximum l with no penalty over the more conventional algorithms.

Our algorithm for the calculation of X_l^m and its θ derivative for all m for a single harmonic degree uses the coupled recurrence

$$\left.\begin{aligned} X_l^{m-1} &= -\left[\frac{d}{d\theta} X_l^m + m \cot\theta X_l^m \right] \Big/ [(l+m)(l-m+1)]^{1/2} \\ &\text{and} \\ \frac{d}{d\theta} X_l^{m-1} &= (m-1)\cot\theta X_l^{m-1} + X_l^m [(l+m)(l-m+1)]^{1/2}, \end{aligned}\right\} \tag{3}$$

with the starting values

$$\left.\begin{aligned} X_l^l &= (-1)^l \left[\frac{2l+1}{4\pi} \frac{1}{(2l)!} \right]^{1/2} \frac{(2l)!}{2^l l!} (\sin\theta)^l \\ &\text{and} \\ \frac{d}{d\theta} X_l^l &= l \cot\theta X_l^l, \end{aligned}\right\} \tag{4}$$

and $(2l-1)!! = 1\cdot 3\cdot 5\ldots(2l-1) = (2l)!/2^l l!$. This recurrence is stable only if we start the recurrence at $m = l$ and go to $m = 0$. We need only compute the X_l^ms for non-negative m, since

$$X_l^{-m} = (-1)^m X_l^m\,.$$

For large l and small θ, the starting value, X_l^l, can underflow (be indistinguishable from zero) on the computer, so we may have to resort to a scaling algorithm. In this case, we start the recurrence with $X_l^l = (-1)^l$ and recur towards $m = 0$, taking care to renormalize X_l^m and its derivative if they get too large. This renormalization is undone at the end of the calculation, allowing large m values to underflow, and the X_l^ms are scaled so that the addition rule is satisfied:

$$\sum_{m=-l}^{l} [X_l^m(\theta)]^2 = \frac{2l+1}{4\pi}\,.$$

In the case of ordinary spherical harmonics, a more elegant way to avoid underflow is to perform the recursions for $W_l^m = (\sin\theta)^{-m} X_l^m$ and then to unscale the W_l^m by powers of $\sin\theta$ at the end (Libbrecht 1985). Eq. (3) becomes

$$\left.\begin{aligned} W_l^{m-1} &= -\sin\theta\left[\frac{d}{d\theta}W_l^m + 2m\cot\theta W_l^m\right]\Big/ [(l+m)(l-m+1)]^{1/2} \\ &\text{and} \\ \frac{d}{d\theta}W_l^{m-1} &= \sin\theta W_l^m[(l+m)(l-m+1)]^{1/2}, \end{aligned}\right\} \quad (5)$$

with starting values

$$\left.\begin{aligned} W_l^l &= (-1)^l\left[\frac{2l+1}{4\pi}\frac{1}{(2l)!}\right]^{1/2}\frac{(2l)!}{2^l l!} \\ &\text{and} \\ \frac{d}{d\theta}W_l^l &= 0. \end{aligned}\right\} \quad (6)$$

Both of the above algorithms work very well and give satisfactory results up to the maximum harmonic degree tested ($l=500$). Libbrecht (1985) discusses possible round-off problems above $l \simeq 200$ but we are unable to generate such problems in practice. It is not clear how to extend the scaled recursion of Libbrecht (1985) to the case of generalized spherical harmonics, which we now discuss in detail.

GENERALIZED SPHERICAL HARMONICS

Since the seminal paper of Phinney & Burridge (1973), much of theoretical global seismology has been developed using expansions of vector and tensor quantities in terms of generalized spherical harmonics. These functions have many advantages, including for example simple behaviour near the pole. The calculation of the generalized spherical harmonics (GSH) proceeds in a similar way to the ordinary spherical harmonics. The GSH are given by

$$Y_l^{N,m} = P_l^{N,m}(\cos\theta)\, e^{im\phi},$$

where the $P_l^{N,m}$ are defined by Phinney & Burridge (1973) and are normalized such that

$$\int_S Y_l^{N,m'*}\, Y_l^{N,m}\, dS = \frac{4\pi}{2l+1}\delta_{mm'}\delta_{ll'}.$$

Phinney & Burridge also give recursion relations for their calculation. We note two of their formulae here:

$$\left.\begin{aligned} \Omega_{N+1}^l Y_l^{N+1,m} + \Omega_N^l Y_l^{N-1,m} &= \sqrt{2}[N\cot\theta - m\,\mathrm{cosec}\,\theta]Y_l^{N,m}, \\ \Omega_N^l Y_l^{N-1,m} - \Omega_{N+1}^l Y_l^{N+1,m} &= \sqrt{2}\frac{d}{d\theta}Y_l^{N,m}, \end{aligned}\right\} \quad (7)$$

where $\Omega_N^l = [(l+N)(l-N+1)/2]^{1/2}$. Note that $\Omega_{-N+1}^l = \Omega_N^l$, so that $\Omega_0^l = \Omega_1^l$, $\Omega_{-1}^l = \Omega_2^l$, etc.

It is possible to write some relationships between the GSH and the ordinary Y_l^ms. Let $\gamma_l = \sqrt{(2l+1)/(4\pi)}$. Then

$$\left.\begin{aligned} &\gamma_l Y_l^{0,m} = Y_l^m, \\ &\frac{1}{\sqrt{2}}\gamma_l\Omega_1^l[Y_l^{-1,m} + Y_l^{+1,m}] = -m\,\mathrm{cosec}\,\theta Y_l^m, \\ &\frac{1}{\sqrt{2}}\gamma_l\Omega_1^l[Y_l^{-1,m} - Y_l^{+1,m}] = \frac{\partial}{\partial\theta}Y_l^m, \\ &\frac{1}{2}\gamma_l\Omega_1^l\Omega_2^l[Y_l^{-2,m} + Y_l^{+2,m}] \\ &\quad = \left[m^2\,\mathrm{cosec}^2\,\theta - \frac{l(l+1)}{2}\right]Y_l^m - \cot\theta\frac{\partial}{\partial\theta}Y_l^m, \\ &\frac{1}{2}\gamma_l\Omega_1^l\Omega_2^l[Y_l^{-2,m} - Y_l^{+2,m}] = m\,\mathrm{cosec}\,\theta\left[\cot\theta Y_l^m - \frac{\partial}{\partial\theta}Y_l^m\right]. \end{aligned}\right\} \quad (8)$$

Relationships for higher N can be found from the recursion formulae above. These relationships are useful if one wants to convert from canonical components back to expressions in r, θ, ϕ involving ordinary spherical harmonics (see Phinney & Burridge 1973) but they are not very useful for the numerical computation of the GSH. To perform the actual computation, we use recurrences analogous to eq. (3) but now we need to be able to recur from both $+l$ and $-l$. The appropriate recursions are as follows.

Downward recursion (from $m=l$):

$$\left.\begin{aligned} P_l^{N,m-1} &= -\left[\frac{d}{d\theta}P_l^{N,m} + (m\cot\theta - N\,\mathrm{cosec}\,\theta)P_l^{N,m}\right]\Big/ [(l+m)(l-m+1)]^{1/2}, \\ \frac{d}{d\theta}P_l^{N,m-1} &= [(m-1)\cot\theta - N\,\mathrm{cosec}\,\theta]P_l^{N,m-1} \\ &\quad + P_l^{N,m}[(l+m)(l-m+1)]^{1/2}, \end{aligned}\right\} \quad (9)$$

with starting values

$$\left.\begin{aligned} P_l^{N,l} &= \left[\frac{(2l)!}{2^l 2^l (l+N)!(l-N)!}\right]^{1/2}(-\sin\theta)^{l-N}(1+\cos\theta)^N, \\ \frac{d}{d\theta}P_l^{N,l} &= [l\cot\theta - N\,\mathrm{cosec}\,\theta]P_l^{N,l}. \end{aligned}\right\} \quad (10)$$

Upward recursion (from $m=-l$):

$$\left.\begin{aligned} P_l^{N,m+1} &= \left[\frac{d}{d\theta}P_l^{N,m} - (m\cot\theta - N\,\mathrm{cosec}\,\theta)P_l^{N,m}\right]\Big/ [(l-m)(l+m+1)]^{1/2}, \\ \frac{d}{d\theta}P_l^{N,m+1} &= -[(m+1)\cot\theta - N\,\mathrm{cosec}\,\theta]P_l^{N,m+1} \\ &\quad - P_l^{N,m}[(l-m)(l+m+1)]^{1/2}, \end{aligned}\right\} \quad (11)$$



Figure 1. Generalized spherical harmonics of harmonic degree 20 for various values of θ. Note that the oscillatory behaviour is centred on $m \simeq N\cos\theta$.

with starting values

$$\left.\begin{aligned} P_l^{N,-l} &= \left[\frac{(2l)!}{2^l 2^l (l+N)!(l-N)!}\right]^{1/2} (\sin\theta)^{l-N}(1-\cos\theta)^N, \\ \frac{d}{d\theta}P_l^{N,-l} &= [l\cot\theta + N\,\mathrm{cosec}\,\theta]P_l^{N,-l}. \end{aligned}\right\} \tag{12}$$

Neither recurrence is stable for the whole range of m (from $-l$ to $+l$) but each is stable where the other isn't and there is a range of m where both are stable. Examination of the governing equation in the large-l approximation shows that the point at which the upward and downward recurrences should meet is $m \simeq N\cos\theta$ (Fig. 1). Again, it is common for the starting values to underflow for large values of l and small values of θ so that a scaling algorithm must be adopted in an analogous fashion to the calculation of the ordinary spherical harmonics. The starting values can then be arbitrary (except they should have the right sign). We use $P_l^{N,-l} = 1$ and $P_l^{N,l} = (-1)^{l-N}$. The final values can be correctly normalized using the addition formula, which, for the normalization of Phinney & Burridge (1973), is

$$\sum_{m=-l}^{l} [P_l^{N,m}(\theta)]^2 = 1\,.$$

Fig. 1 shows $P_{20}^{N,m}$ for various values of θ. The algorithm has been compared for accuracy with a conventional calculation using upward recursion of Jacobi polynomials (Szegö 1939). The two algorithms agree to a part in 10^{12} using 64-bit arithmetic up to a harmonic degree of 256. We have also compared the algorithm with one published in Doornbos (1988) based on the results of Backus (1964). This latter algorithm peforms quite poorly at values of θ close to 90° and precision falls dramatically above about $l = 50$.

ROTATION OF SPHERICAL HARMONICS

One immediate special application of GSH is in the computation of the rotation matrix for spherical harmonics. Central to this calculation is the computation of the GSH at $\theta = 90°$ (Edmonds 1960, Chapter 4). In the notation of Edmonds,

$$d^l_{mm'}(\pi/2) = (-1)^{m'+m} P_l^{m',m}(\pi/2)\,, \tag{13}$$

which is the same as the $T_l^{m,j}$ matrix of Goldstein (1984). A particularly efficient form of the algorithm can be developed for this special case since underflow is not a problem and the recurrence is to $m = 0$. Since all values of N are to be computed, the following symmetries can be used to reduce calculation:

$$P_l^{-N,-m} = (-1)^{m+N} P_l^{N,m} \quad \text{and} \quad P_l^{m,N} = (-1)^{N+m} P_l^{N,m}, \tag{14}$$

which are valid for all θ, and

$$P_l^{-N,m} = (-1)^{l+m} P_l^{N,m} \quad \text{and} \quad P_l^{N,-m} = (-1)^{l+N} P_l^{N,m}, \tag{15}$$

which are valid only at $\theta = 90°$. This algorithm proves to be about twice as fast as the Goldstein algorithm and gives results which agree to a part in 10^{15} up to a harmonic degree of 256.

FORTRAN and C subroutines that implement these algorithms are available over the Internet from the authors (contact gmasters@ucsd.edu for details).

ACKNOWLEDGMENTS

We would like to thank J. Tromp for a helpful review and, in particular, for pointing us to the Libbrecht paper. This research was supported by grants from the National Science Foundation.

REFERENCES

Backus, G.E., 1964. Geographical interpretation of measurements of average phase velocities of surface waves over great circular and great semi-circular paths, *Bull. seism. Soc. Am.*, **54**, 571–610.

Doornbos, D., 1988. *Seismological Algorithms: Computational Methods and Computer Programs,* Academic Press, London.

Edmonds, A.R., 1960. *Angular Momentum and Quantum Mechanics,* Princeton University Press, Princeton, NJ.

Goldstein, J.D., 1984. The effect of coordinate system rotations on spherical harmonic expansions: a numerical method, *J. geophys. Res.*, **89**, 4413–4418.

Libbrecht, K.G., 1985. Practical considerations for the generation of large-order spherical harmonics, *Solar Phys.*, **99**, 371–373.

Phinney, R.A. & Burridge, R., 1973. Representation of the elastic-gravitational excitation of a spherical-earth model by generalized spherical harmonics, *Geophys. J. R. astr. Soc.*, **34**, 451–487.

Press, W.H., Teukolsky, S.A., Vetterling, W.T. & Flannery, B.P., 1992. *Numerical Recipes. The Art of Scientific Computing,* 2nd edn, Cambridge University Press, New York.

Szegö, G., 1939. *Orthogonal Polynomials,* Am. Math. Soc., Colloq. Publications, Vol. 23, New York.